

# Verified Profunctor Optics in Idris

Oliver Balfour

October 28, 2021

## **Abstract**

Optics are a commonly used design pattern in industrial functional programming. They are convenient combinators for reading and updating fields in composite data structures. We discuss profunctor optics, a modern formulation of optics which is more flexible than the more common van Laarhoven formulation. This report discusses the implementation and formal verification of profunctor optics in Idris, a dependently typed functional programming language and theorem prover.

# Contents

Introduction . . . . .	2
Background . . . . .	3
Idris . . . . .	3
Dependent Types . . . . .	3
Propositions as Types . . . . .	4
Proof Techniques . . . . .	5
Limitations . . . . .	6
Functors . . . . .	7
Profunctors . . . . .	8
Optics . . . . .	9
Profunctor Optics . . . . .	11
Formally Verified Profunctor Optics . . . . .	13
Related Work . . . . .	13
Conclusion . . . . .	13
Appendix: Source Code . . . . .	16
Morphisms: Morphism.idr . . . . .	16
Verified Functors and Applicatives: VFunctor.idr . . . . .	17
Verified Profunctors: VProfunctor.idr . . . . .	24
Simple Optics: PrimitiveOptics.idr . . . . .	27
van Laarhoven Optics: LaarhovenOptics.idr . . . . .	29
Profunctor Optics: Main.idr . . . . .	30

## Introduction

The view-update problem is the problem of how to neatly read and write small components of large composite data structures (Foster et al. 2005). In imperative languages, objects are generally mutated in-place, circumventing the view-update problem altogether. Pure functional programming languages however are not afforded mutable variables, making the issue pernicious in industrial programs with highly complex data structures.

Optics are a pure functional solution to the view-update problem. Data structures representing components in real world systems frequently have dozens of fields and nested data structures with additional complexity. In a pure functional language, updating a field in a composite data type such as `Maybe (a, Bool)` requires boilerplate functions for every such composite type as in the below Idris code:

```
updateComplexType : (a -> b) -> Maybe (a, Bool) -> Maybe (b, Bool)
updateComplexType f (Just (x, y)) = Just (f x, y)
updateComplexType f Nothing = Nothing
```

As data structures become more complex, writing getters and setters becomes a tedious and bug-prone task. Optics are objects which represent a view into a data type which can be composed to create views into composite types, and used to view or update fields. Using the profunctor optics library discussed in this report, the above `updateComplexType` function may be defined as `updateComplexType = update (op .  $\pi_1$ )` where `op` is an optic for optional (`Maybe a`) types and  $\pi_1$  is a left projection optic for product types. Profunctor optics are a very flexible and powerful encoding of optics, however they are highly complex, demonstrating a need for quality assurance.

Even in imperative languages there are often many benefits from using immutable objects. In JavaScript for example, there is an increasing trend towards pure functional state management for designing user interfaces, termed *declarative UI* (Steinberger 2021). Libraries such as Redux.js (Abramov et al. 2015) use an immutable state object with a group of actions that act on the state type whenever an event is triggered by user interactions. This presents numerous benefits such as simple control flow and undo/redo functionality. However, this requires a new state object after each event with perhaps a single field changed. The conventional approach in JavaScript is to use Immer.js (Weststrate et al. 2019), which rather than using pure optics exploits esoteric language features to emulate mutability on immutable objects. However, there is no fundamental reason why optics would not work equally well.

In statically typed languages, types correspond with certain logical propositions and programs serve as proofs of those propositions (Wadler 2015). This insight is known as the Curry-Howard correspondence (Sørensen and Urzyczyn 2006) and it underpins theorem provers and formal verification. Dependent types are a feature of some type systems which allows types to depend on values. Dependent types allow programmers to encode first order logical propositions and equalities between expressions into the type system and prove many useful theorems and properties of their programs.

Idris is a dependently-typed functional programming language similar to Haskell which may be used as a theorem prover. This report discusses using Idris to both implement and formally verify a profunctor optics library. Dependent types are used to express and prove that the profunctor optics adhere to all relevant mathematical laws and desirable properties.

## Background

### Idris

Idris (Brady 2013) is a Haskell-like functional programming language with first-class support for dependent types. It is an actively developed experimental research language. Syntactically Idris and Haskell are almost identical, the most notable difference is that `:` is used to declare types and `::` is the list `cons` constructor. Additionally, types are first class citizens so functions may accept or return types (values may depend on types), a strict generalisation of Haskell which only allows types to depend on types (type constructors). Idris also has implicit arguments denoted with `{x : a}` syntax which are inferred from context.

Idris additionally has linear types based on quantitative type theory (Atkey 2018) which allow types to be annotated with requirements that they must be used exactly 0 or 1 times at runtime (Brady 2021). Idris also has implicit (inferred) arguments. Unlike Haskell, Idris does not possess type inference, as type inference is undecidable in general for dependent types with non-empty typing contexts (Dowek 1993).

Idris is unique in that it is a practical and simple functional programming language to understand given prerequisite Haskell experience, and it doubles up as a theorem prover. The type system is powerful enough to encode theorems about equalities between expressions and universal and existential quantifiers. This allows programmers to express and prove complex properties and invariants of their programs alongside their code, which makes languages like Idris a good candidate language for critical infrastructure and similar systems.

### Dependent Types

Dependent types are types that depend on values. For example, the Idris type `Vect 3 Int` is inhabited by vectors of precisely 3 integers. We say the type is indexed by the value 3.

Some other languages have equivalent types such as `std::array<int, 3>` in C++. However, in C++, non-type template parameters (that is, values the type depends on) must be statically evaluated because generic types are monomorphised at compile time (ISO 2020, 14.1.4). This means template arguments cannot be non-trivial expressions as in Idris.

There are two main kinds of dependent types.  $\Pi$  types generalise the `Vect 3 Int` example above. The type  $\Pi x.Px$ , which is expressed as `(x:a) -> P x` in Idris for some `P : (x:a) -> Type` is a function type where the codomain type depends on the value of the argument `x`. This allows functions to dynamically compute their return types in a type-safe manner. For instance, the

`replicate` function in the Idris standard library has the type `replicate : (len : Nat) -> a -> Vect len a`, using a  $\Pi$  type to construct a length `len` vector of copies of an object.

The other kind is  $\Sigma$  types, which in Idris are known as dependent pairs. The type  $\Sigma x.Px$  corresponds with the dependent pair  $(x:a ** P\ x)$  which is a pair of a value and a type where the type may depend on the value.

As types can depend on values, Idris has an equality type `=` indexed by two values. It has one constructor `Refl : x = x` (reflexivity). An instance of `Refl : a = b` in some cases is obtainable using type rewriting rules discussed later, in which case the expressions `a` and `b` share the same normal form and are intensionally equal.

Dependent types are useful because they allow programmers to express more sophisticated types such as length indexed vectors, which allow programmers to write total matrix multiplication functions. Additionally, logical propositions correspond with types, and dependent types are expressive enough to allow a language to be used as a theorem prover and formally verify properties of programs.

## Propositions as Types

The Curry-Howard correspondence, also known as *Propositions as Types*, is the observation that propositions in a logic correspond with types in a language and proofs correspond with function definitions (Wadler 2015). This observation underpins theorem provers like Idris, Coq<sup>1</sup> and Agda<sup>2</sup>. The theorem statement or goal is encoded in a type signature. The function body is a proof of the goal. If the program is well-typed, the proof is correct.

Dependent types are expressive enough that they can encode an intuitionistic or constructive logic complete with implications, conjunction, disjunction, negation, quantifiers and equalities. Intuitionistic logics are similar to classical logic except they do not have the law of the excluded middle  $p \vee \neg p$  or double negation  $\neg\neg p \Leftrightarrow p$  (Moschovakis 1999). Additionally, existence statements must have an explicit witness.

In Idris, the type `a` is interpreted as a proposition  $a$ , where  $a$  is true if `a` as a type is inhabited. A proof of  $a$  is simply an object of type `a`. The function type `a -> b` is interpreted as a logical implication  $a \implies b$ . Intuitively, if a total function of type `a -> b` exists then the existence of an `a` guarantees the existence of a `b`. Logical negation is encoded as `a -> Void` where `Void` is uninhabited.

The equality type is especially useful in conjunction with. If  $a = b$  then `a = b` is inhabited by all of the constructive proofs of this equality, and if no proof exists it is uninhabited and thus false.

A list of corresponding logical connectives is tabulated below.  $\Sigma$  types are encoded using a construct called dependent pairs, which is not discussed in this report.  $\Pi$  types are encoded with function types where the return type depends on the argument.

---

<sup>1</sup><https://coq.inria.fr/>

<sup>2</sup><https://wiki.portal.chalmers.se/agda/pmwiki.php>

Logic	Type Theory	Idris Type
$T$	$\top$	<code>()</code>
$F$	$\perp$	<code>Void</code>
$a \wedge b$	$a \times b$	<code>(a, b)</code>
$a \vee b$	$a + b$	<code>Either a b</code>
$a \Rightarrow b$	$a \rightarrow b$	<code>a -&gt; b</code>
$\forall x.Px$	$\Pi x.Px$	<code>(x:a) -&gt; P x</code>
$\exists x.Px$	$\Sigma x.Px$	<code>(x:a ** P x)</code>
$\neg p$	$p \rightarrow \perp$	<code>p -&gt; Void</code>
$a = b$	$a = b$	<code>a = b</code>

Table 1: Corresponding connectives and quantifiers. Note that the predicates in Idris are of the form  $P : (x : a) \rightarrow \text{Type}$  where  $P\ x = ()$  or  $P\ x = \text{Void}$ .

## Proof Techniques

Idris will reduce values in types to their normal form by applying function definitions. It will attempt to unify both sides of equality types as well by reducing either side until it coincides with the other. This allows proofs to skip many intermediate simplification steps. Idris will generally reduce values in types to their normal form, analogous to simplifying mathematical expressions. For instance `3 + 7 = 11` will be rewritten to `10 = 11` (which of course is uninhabited).

This allows us to write simple proofs as below, which are analogous to unit tests.

```
fact : Nat -> Nat
fact Z = 1
fact (S n) = (S n) * fact n

factTheorem : fact 5 = 5 * 4 * 3 * 2 * 1
factTheorem = Refl

factTheorem2 : fact 5 = 120
factTheorem2 = Refl

factTheorem2 : (S n) * fact n = fact (S n)
factTheorem2 = Refl
```

The main proof techniques in Idris are structural induction, rewriting types and ex falso quodlibet.

Structural induction is the most common tool. This entails case splitting a theorem over each constructor and recursively invoking the theorem on smaller components of an inductively defined structure. If Idris can determine the theorem is total as the recursive calls eventually reach the base case, the proof will type check. Recursive calls are analogous to inductive hypotheses.

For example,

```
-- ∀ n : Nat. n + 0 = n
natPlusZeroId : (n : Nat) -> n + 0 = n
natPlusZeroId z = Refl
natPlusZeroId (S n) = cong S (natPlusZeroId n)

-- ∀ xs : List a. xs ++ [] = xs
listConcatRightNilId : (xs : List a) -> xs ++ [] = xs
listConcatRightNilId [] = Refl
listConcatRightNilId (x::xs) = cong (x::) (listConcatRightNilId xs)
```

These proofs invoke a lemma in the Idris Prelude, `cong : (f:t->u) -> (a = b) -> (f a = f b)`, which is analogous to the rule  $\forall f. a = b \implies f(a) = f(b)$  in mathematics.

Idris also provides a facility for rewriting the goal type using an equality. For example:

```
trans' : a = b -> b = c -> a = c
trans' p1 p2 =
  -- goal: a = c
  rewrite p1 in -- replace `a` with `b` in `a = c`
  -- new goal: b = c
  p2
```

Rewriting can be convenient, however using a number of rewrites makes proofs difficult to follow. Prelude functions such as `trans`, `sym`, `cong` and `replace` can accomplish the same tasks with a more conventional proof structure.

As intuitionistic logics do not have the law of the excluded middle or double negation, proof by contradiction is not possible. Instead, *ex falso quodlibet*, the principle of explosion, must be used. In some cases a function has cases which are not possible but well-typed proofs must exist for those cases to satisfy the totality checker. In this case, rather than deriving a contradiction to show the state is not possible, the contradiction can be used with the function `void : Void -> a` to derive the proof goal.

Idris has holes like Haskell, which are placeholder expressions denoted `?hole_name`. There is a `:t hole_name` command in the Idris REPL which prints out the typing context and goal, much like other theorem provers like Coq. This is immensely useful in developing proofs.

## Limitations

Dependently typed theorem provers are intuitionistic in nature, which is strictly less powerful than classical logic. There exist theorems which can be proven with classical logic for which no constructive proof in Idris exists.

Double negation cancellation is not true in general, as there is no canonical map `((a -> Void) -> Void) -> a`. Existence statements cannot be proven without finding a witness to the proof,



so a proof by contradiction that  $\neg\forall x.P(x)$  does not imply  $\exists x.\neg P(x)$ . Instead, a dependent pair containing an explicit  $x$  satisfying  $\neg P(x)$  must be constructed, which may not be possible.

Additionally, there is a distinction between intensional and extensional equality of functions (nLab 2021a). In mathematics, the statements  $f = g$  and  $\forall x.f(x) = g(x)$  are equivalent. In Idris however, only the forward implication is true. Function equality is intensional, meaning functions are equal if and only if the normal form of their lambda expressions are  $\alpha$ -equivalent, so they are the equal up to renaming bound variables. In many cases extensionally equal functions are not intensionally equal, so the Idris equality type may not be helpful. It is possible to use the built-in `believe_me : a -> b` proof to introduce an extensionality axiom, however Idris cannot rewrite types if they invoke axioms as there essentially is no definition to substitute.

These limitations mean that many theorems of interest either cannot be proven or are much more difficult to prove in Idris than in a classical logic.

## Functors

Before discussing profunctors, we discuss categories, functors, applicative functors and monads. A category is a mathematical object which consists of a collection of objects and between any two objects a collection of arrows or morphisms (Mac Lane 1970). We will only discuss locally small categories so we may assume these collections of morphisms are sets, called Hom-sets. The only properties categories must have is an associative composition operation on morphisms and an identity morphism on each object. Categories are a useful abstraction as they generalise objects and structure preserving maps between them from many different fields. There is a category of sets where objects are sets and Hom-sets contain functions,  $\text{Hom}(A, B) = \{f : A \rightarrow B \text{ is a function}\}$ . Morphism composition is function composition, and there exists an identity function on each set. In group theory, there is a category of groups where objects are groups and morphisms are group homomorphisms. Many other examples exist, for example partially ordered sets form categories where objects are elements and exactly one morphism exists between every ordered pair.

Notably, types and total functions in Idris form a category similar to the category of sets. For convenience, these categories are assumed the same.

Functors are structure preserving maps between categories (and thus morphisms in the category of categories). They consist of two components mapping objects and morphisms from the domain category to objects and morphisms in the codomain category. Functors respect identities  $F(\text{id}_X) = \text{id}_{F(X)}$  and composition  $F(f \circ g) = F(f) \circ F(g)$ . An endofunctor is a functor which maps into the same category.

In Idris, generic containers such as lists and trees are endofunctors. The type constructor `List : Type -> Type` is the component of the functor mapping objects, and the `map : (a -> b) -> (List a -> List b)` function is the component mapping morphisms. Additionally, the partially applied arrow type `a->` is a functor, the covariant Hom functor (and partially applied Hom profunctor).

Monads are a subset of endofunctors equipped with two maps  $\eta : a \rightarrow m\ a$  and  $\mu : m\ (m\ a) \rightarrow m\ a$  named `pure`/`return` and `join` respectively, where `m` is the monad. In functional programming, they can be used to encapsulate and compose side effecting functions in a type safe way (Moggi 1991). The `IO` type constructor is a monad representing side effects which allows side effects to be composed and enforce that functions emitting side effects have a return type containing `IO`. Lists form a monad where `pure x = [x]` and `join = concat` and function composition results in a list of all possible applications of functions to arguments.

Monads satisfy the following laws:

1.  $\forall f, x. (\text{return } \text{'join'}\ f)\ x = f\ x$  (left identity law, tildes denote infix function calls)
2.  $\forall f, x. (f\ \text{'join'}\ \text{return})\ x = f\ x$  (right identity law)
3.  $\forall f, g, h, x. f\ \text{'join'}\ (g\ \text{'join'}\ h)\ \$\ x = (f\ \text{'join'}\ g)\ \text{'join'}\ h\ \$\ x$  (associativity)

Monads naturally give rise to a category known as a Kleisli category (Mac Lane 1970 p.147). In this category, the objects are the same as the category of types, and the morphisms are each of the form  $a \rightarrow m\ b$  so  $\text{Hom}(a, b) = \{f : a \rightarrow m\ b\}$ . Morphism composition utilises the `join` operation, so  $f \circ_m g = \mu \circ \text{fmap}(f) \circ g$ .

Applicative functors are a subset of endofunctors introduced by McBride and Paterson 2008 as a useful abstraction in functional programming intermediate between endofunctors and monads. They are endofunctors equipped with maps `pure : a -> f a` and `ap : f (a -> b) -> (f a -> f b)` (written `<*>` as an infix operator) for every applicative `f` satisfying the following properties

1.  $\forall v. \text{pure id } \text{<*>} v = v$  (identity law)
2.  $\forall g, x. \text{pure g } \text{<*>} \text{pure x} = \text{pure (g x)}$  (homomorphism law)
3.  $\forall u, y. u\ \text{<*>} \text{pure y} = \text{pure } (\backslash x \Rightarrow x\ y)\ \text{<*>} y$  (interchange law)
4.  $\forall u, v, w. ((\text{pure } \text{'.'})\ \text{<*>} u)\ \text{<*>} v)\ \text{<*>} w = u\ \text{<*>} (v\ \text{<*>} w)$  (composition law)

## Profunctors

Profunctors are a generalisation of functors which relate to Hom-sets. A profunctor from category  $C$  to  $D$  formally is a functor  $D^{op} \times C \rightarrow \mathbf{Set}$ , where  $D^{op}$  is the dual category of  $D$  and  $\times$  is similar to the Cartesian product (nLab 2021b). The dual category has the same objects as  $D$ , but the directions of all morphisms are reversed.

As we assume the categories of types and sets are identical, a profunctor in Idris is a type constructor `p` which takes two type variables equipped with a map `dimap : (a -> b) -> (c -> d) -> p b c -> p a d`. Then every profunctor  $p$  gives rise to a (covariant) functor  $p(a, -)$  for all  $a$ , and a contravariant functor (functor mapping from the dual category)  $p(-, a)$ . Profunctors must

adhere to the identity and composition laws for functors, which are expressed in Idris as `(x : a) -> dimap id id x = x` and `(x : a) -> dimap (f' . f) (g . g') x = dimap f g . dimap f' g'`. These laws are expressed as extensional equalities as in many cases it is not possible to constructively prove these functions are intensionally equal. This is because without an explicit `x` argument structural induction is not possible and Idris may not be able to equate the lambda expressions directly.

In Idris, profunctors usually correspond to arrow-like types. For instance, `->` is a profunctor. The `Hom` function is a profunctor, and the `Hom` functor is simply the partially applied `Hom` profunctor. Additionally, for any monad `m` the `Hom` profunctor in its Kleisli category is a profunctor in the category of types. Hash maps or dictionaries form profunctors (Milewski 2017a). Several other profunctors discussed later underpin the construction of profunctor optics.

Cartesian profunctors are profunctors equipped with functions `first : p a b -> p (a,c) (b,c)` and `second`. Cocartesian profunctors have functions `left : p a b -> p (Either a c) (Either b c)` and `right`.

## Optics

Optics are data accessors that ease reading and writing into composite data structures. Industrial programs tend to have very complex and deeply nested data structures, which makes tasks like copying and updating a single field in a deeply nested data structure in a functional style very cumbersome. Optics are an elegant solution to this problem. They are objects which represent a view into a field of a data structure which can be composed for nested structures and used to view and update the field they model.

There are many encodings of optics. This report discusses simple algebraic data type optics, van Laarhoven optics (Laarhoven 2011) and profunctor optics (Pickering, Gibbons, and Wu 2017). Most established implementations such as the `lens` library in Haskell (Kmett 2012) use the van Laarhoven design, however these have many limitations. One of the principal benefits of optics is that they compose elegantly, however the van Laarhoven encoding makes a distinction between lenses (optics for product types) and prisms (optics for sum types) and does not allow composition of lenses and optics, so you cannot express an optic for the integer in a `Maybe (Integer, String)` type (Pickering, Gibbons, and Wu 2017).

However, profunctor optics generalise optics to work around these issues. As with many other functional programming design patterns, they are inspired by category theory, specifically the notion of a profunctor. Profunctor optics are generic over the typeclass of profunctors, so they allow choice of profunctor in which to use an optic. This allows programmers to not just use optics to view and update, but to also accumulate side effects and recover constructors for sum types in the process.

Formal verification of profunctor optics is a natural application of dependent types and propositions as types as optics are very complex and warrant quality assurance measures.

The simplest encoding of optics is to create typeclasses (interfaces in Idris) for lenses, prisms, traversals and adapters. Lenses are optics for product types that allow viewing and updating fields. Prisms are optics for sum types that allow pattern matching if a field is present and constructing a sum type from one of the components. Adapters and traversals are optics for isomorphic types and container types respectively and are not discussed in this report.

In the below encoding, `PrimitiveLens a b (a,c) (b,c)` means a view into the `a` in `(a,c)`. The other two type variables add a degree of freedom when updating tuples to change the type of the left element of the tuple.

```
record PrimitiveLens a b s t where
  constructor MkPrimLens
  view : s -> a
  update : (b, s) -> t
```

```
record PrimitivePrism a b s t where
  constructor MkPrimLens
  match : s -> Either t a
  build : b -> t
```

```
-- Left projection lens
π1 : PrimitiveLens a b (a,c) (b,c)
π1 = MkPrimLens fst update where
  update : (b, (a, c)) -> (b, c)
  update (x', (x, y)) = (x', y)
```

Then `view π1 (2, True) == 2`. However, there is no clear way to compose two lenses or two prisms using this encoding, and it is not possible to compose a lens and a prism using these two typeclasses.

A more powerful encoding is van Laarhoven functor transformer lenses (Laarhoven 2011). These are parameterised over the functor typeclass, where different functors applied to the optics change how they behave. Note that `VFunctor` is the interface for verified functors in the source code.

```
LaarhovenLens : {f : Type -> Type} -> VFunctor f => Type -> Type -> Type
LaarhovenLens a s = (a -> f a) -> (s -> f s)
```

```
-- Left projection example
laarhovenProj : {f : Type -> Type} -> VFunctor f =>
  LaarhovenLens {f=f} a (a,b)
laarhovenProj g (x, y) = fmap (,y) (g x)
```

Using the `Const a` functor a getter is produced. The `Const a b` type stores a value of type `a` and ignores the second type variable. By using `f=Const a` in a lens `LaarhovenLens a s` we get a function `(a -> Const a a) -> (s -> Const a s)`. `Const a a` contains an `a` so the

function `MkConst` can be passed to this function to get a function `s -> Const a s`, and the return type here contains an `a` as desired, so this is a getter. Likewise, using the identity functor it is clear the resulting function can be used to update the field as below:

```
view : LaarhovenLens {f=Const a} a s -> (s -> a)
view optic structure = unConst $
  (optic (\x => MkConst x) structure)

update : LaarhovenLens {f=Id} a s -> ((a, s) -> s)
update optic (field, structure) = unId $
  (optic (\x => MkId field) structure)
```

These optics are simple functions and so support composition, however lenses and prisms are still mutually exclusive and cannot be composed. This leaves profunctor optics, which generalise van Laarhoven's functor transformer lenses and are flexible enough to support composition.

## Profunctor Optics

Profunctor optics are functions of the type `VProfunctor p => p a b -> p s t`. The type variables have the same naming schema as above, so `a` is the type of the field the optic focuses on in a composite data structure of type `s`. Much like van Laarhoven optics, these optics have different behaviours when different profunctors are chosen.

Lenses are optics which are defined only for Cartesian profunctors, which have a function `first : p a b -> p (a,c) (b,c)` and thus preserve product types. Dually, prisms are optics which are defined only for Cocartesian profunctors, which preserve sum types. The following type aliases are used for convenience:

```
Optic p a b s t = p a b -> p s t

Lens a b s t = {p : Type -> Type -> Type}
  -> Cartesian p => Optic p a b s t

Prism a b s t = {p : Type -> Type -> Type}
  -> Cocartesian p => Optic p a b s t
```

Profunctor optics can be composed freely: a lens composed with a prism is simply an optic parameterised over Cartesian and Cocartesian profunctors. This makes them the most powerful encoding of optics.

Profunctor optics for `Either` and `(,)` types are exactly the functions `left/right` and `first/second`. However, profunctor optics for more complex data structures are much more difficult to write explicitly. For instance, the prism on optional values is

```
op : Prism a b (Maybe a) (Maybe b)
op = dimap (maybe (Left Nothing) Right) (either id Just) . right
```

However, there is a proven correspondence between the simple, concrete algebraic data type optics and profunctor optics (Boisseau and Gibbons 2018). This means optics can be written using the concrete representation and then converted to profunctor optics. As an example, the following code ported from the Haskell code in Pickering, Gibbons, and Wu 2017 converts a concrete prism to a profunctor prism:

```
prismPrimToPro : PrimitivePrism a b s t -> Prism a b s t
prismPrimToPro (MkPrimPrism m b) = dimap m (either id b) . right
```

Different profunctors make profunctor optics exhibit different behaviours. The Hom profunctor allows for updating fields in composite types. It turns optics into functions of type  $(a \rightarrow b) \rightarrow (s \rightarrow t)$ , which uses a function  $a \rightarrow b$  to replace the  $a$  in an  $s$  with a  $b$  to get a  $t$ . The Hom profunctor in the Kleisli category of any monad does the same while accumulating side effects.

The Forget profunctor turns optics into getters. It is Cartesian and not Cocartesian, so it only works for lenses.

```
record Forget r a b where
  constructor MkForget -- MkForget : (a->r) -> Forget r a b
  unForget : a -> r      -- unForget : Forget r a b -> (a->r)
```

```
VProfunctor (Forget r) where ...
```

```
-- Using the `Forget a` profunctor we can wrap and pass the identity
-- function to get a wrapped getter
```

```
 $\pi_1$  {p=Forget a} : Forget a a b -> Forget a (a, c) (b, c)
unForget ( $\pi_1$  {p=Forget a} (MkForget (\x => x))) : (a, c) -> a
```

```
-- More generally
```

```
view : {a : Type} -> Lens a b s t -> s -> a
view optic x = unForget (optic {p=Forget a} (MkForget (\x => x))) x
```

The Const profunctor recovers sum type constructors. It is only Cocartesian so it only works for prisms.

```
record Const r a where
  constructor MkConst -- MkConst : a -> Const r a
  unConst : a          -- unConst : Const r a -> a
```

```
VProfunctor Const where ...
```

```
build : Prism a b s t -> b -> t
build optic x = unConst (optic {p=Const} (MkConst x))
```

## Formally Verified Profunctor Optics

Profunctor optics are both highly complex and highly useful. As a result, a formally verified implementation of profunctor optics is desirable. A small profunctor optics library was developed in Idris, with source code in the appendix. This library includes profunctor lenses, prisms and traversals, which are optics for traversing containers like lists and trees. It includes verified functor, applicative functor and profunctor interfaces and proofs that optics behave correctly. Propositions as types is used to encode laws into these interfaces, and the aforementioned proof techniques are used to verify these laws.

Function equality in Idris is intensional. Some profunctors such as `Forget r` contain functions and others such as `Const` contain constants. To verify the profunctor law `dimap id id x = x` where `x` is a function, a constructive proof of intensional equality is required. This is not always possible, so an extensionality axiom `extensionality : {f, g : a -> b} -> ((x : a) -> f x = g x) -> f = g` was introduced. This is used most notably in the proof that the Kleisli Hom profunctor is a profunctor.

## Related Work

Much existing work is focused on the correspondence between the concrete and profunctor representations of lenses, prisms, adapters and traversals. Boisseau and Gibbons 2018 provides a proof of the correspondence with the Yoneda lemma. Other work by Milewski 2017b, Román 2020 and Boisseau 2018 provide proofs invoking more complex machinery such as Tambara modules and name the correspondence the profunctor optic representation theorem. Román 2020 also wrote a partial proof of the profunctor representation theorem in Agda <sup>3</sup>. Additional work has been done on codifying lawfulness in optics in Riley 2018.

Future research on profunctor optics for dependent types such as type indexed syntax trees could be very useful. This would enable programming languages written in Idris to use dependent types to verify all syntax trees are well typed and use optics to elegantly traverse, view and update subtrees. Additionally, a complete verified library of profunctor optics would be very useful.

## Conclusion

We have constructed a practical profunctor optics library along with formal verification of many of its components. There are many more components which could be verified, such as adding laws to the `Cartesian` and `Cocartesian` interfaces and the verifying the methods for converting concrete optics into profunctor optics produce identically behaving optics. Future work on optics for dependently typed data structures is warranted.

---

<sup>3</sup><https://github.com/mroman42/vitrea-agda>

# Bibliography

- Abramov, Dan et al. (2015). *Redux: A Predictable State Container for JS Apps*. URL: <https://redux.js.org/> (visited on 10/26/2021).
- Atkey, Robert (2018). “Syntax and semantics of quantitative type theory”. In: *Proceedings of the 33rd Annual ACM/IEEE Symposium on Logic in Computer Science*, pp. 56–65.
- Boisseau, Guillaume (2018). “Understanding profunctor optics: a representation theorem”. In: *Conference Companion of the 2nd International Conference on Art, Science, and Engineering of Programming*, pp. 30–32.
- Boisseau, Guillaume and Jeremy Gibbons (2018). “What you needa know about Yoneda: Profunctor optics and the Yoneda Lemma (functional pearl)”. In: *Proceedings of the ACM on Programming Languages* 2.ICFP, pp. 1–27.
- Brady, Edwin (2013). “Idris, a general-purpose dependently typed programming language: Design and implementation”. In: *Journal of functional programming* 23.5, pp. 552–593.
- (2021). “Idris 2: Quantitative Type Theory in Practice”. In: *arXiv preprint arXiv:2104.00480*.
- Dowek, Gilles (1993). “The undecidability of typability in the  $\lambda\Pi$ -calculus”. In: *International Conference on Typed Lambda Calculi and Applications*. Springer, pp. 139–145.
- Foster, J Nathan et al. (2005). “Combinators for bi-directional tree transformations: a linguistic approach to the view update problem”. In: *ACM SIGPLAN Notices* 40.1, pp. 233–246.
- ISO (Feb. 2020). *ISO/IEC 14882:2020 Information technology — Programming languages — C++*. Geneva, Switzerland: International Organization for Standardization. URL: <https://www.iso.org/standard/79358.html>.
- Kmett, Edward (2012). *Lens: Lenses, Folds, and Traversals*. URL: <https://lens.github.io/> (visited on 10/26/2021).
- Laarhoven, Twan van (2011). “Lenses: viewing and updating data structures in Haskell”. Institute for Computing and Information Sciences, Radboud University Nijmegen. URL: <https://www.twanvl.nl/blog/news/2011-05-19-lenses-talk>.
- Mac Lane, Saunders (1970). *Categories for the working mathematician*. 2nd ed. Springer-Verlag New York.
- McBride, Conor and Ross Paterson (2008). “Applicative programming with effects”. In: *Journal of functional programming* 18.1, pp. 1–13.
- Milewski, Bartosz (2017a). *Ends and Coends*. URL: <https://bartoszmilewski.com/2017/03/29/ends-and-coends/> (visited on 10/26/2021).



- Milewski, Bartosz (2017b). *Profunctor Optics: The Categorical View*. URL: <https://bartoszmilewski.com/2017/07/07/profunctor-optics-the-categorical-view/> (visited on 08/07/2021).
- Moggi, Eugenio (1991). “Notions of computation and monads”. In: *Information and computation* 93.1, pp. 55–92.
- Moschovakis, Joan (1999). “Intuitionistic logic”. In.
- nLab (2021a). *Function extensionality*. <http://ncatlab.org/nlab/show/function+extensionality>. revision 11. (Visited on 10/28/2021).
- (2021b). *Profunctor*. <http://ncatlab.org/nlab/show/profunctor>. revision 71. (Visited on 10/26/2021).
- Pickering, Matthew, Jeremy Gibbons, and Nicolas Wu (2017). “Profunctor optics: Modular data accessors”. In: *arXiv preprint arXiv:1703.10857*.
- Riley, Mitchell (2018). “Categories of optics”. In: *arXiv preprint arXiv:1809.00738*.
- Román, Mario (2020). “Profunctor optics and traversals”. In: *arXiv preprint arXiv:2001.08045*.
- Sørensen, Morten Heine and Pawel Urzyczyn (2006). *Lectures on the Curry-Howard isomorphism*. Elsevier.
- Steinberger, Peter (2021). *The Shift to Declarative UI*. URL: <https://increment.com/mobile/the-shift-to-declarative-ui/> (visited on 10/26/2021).
- Wadler, Philip (2015). “Propositions as types”. In: *Communications of the ACM* 58.12, pp. 75–84.
- Weststrate, Michel et al. (2019). *Immer: Create the next immutable state tree by simply modifying the current tree*. URL: <https://github.com/immerjs/immer> (visited on 10/26/2021).

## Appendix: Source Code

Mirrored at <https://github.com/OliverBalfour/ProfunctorOptics>

### Morphisms: Morphism.idr

```
1  module Category.Morphism
2
3  %default total
4
5  -- Derived from Data.Morphisms
6
7  -- Morphisms in the category of Idris types
8  -- This wrapper exists to help Idris unify types in some proofs
9  public export
10 record Morphism a b where
11   constructor Mor
12   applyMor : a -> b
13
14 infixr 1 ~>
15
16 public export
17 (~>) : Type -> Type -> Type
18 (~>) = Morphism
19
20 -- Morphisms in a Kleisli category
21 -- Functions of type a -> f b for a functor or monad f
22 public export
23 record KleisliMorphism (f : Type -> Type) a b where
24   constructor Kleisli
25   applyKleisli : a -> f b
26
27 infixr 1 ~~>
28
29 public export
30 (~~>) : {f : Type -> Type} -> Type -> Type -> Type
31 (~~>) = KleisliMorphism f
32
33 -- Helpers
34
35 public export
36 eta : (a -> b) -> (a -> b)
37 eta f = \x => f x
38
39 -- f = \x => f x
40 public export
41 ext : (f : a -> b) -> (eta f = f)
```

```

42 ext f = Refl
43
44 -- id . f = f
45 public export
46 idCompLeftId : (f : a -> b) -> (\x => x) . f = f
47 idCompLeftId f = ext f
48
49 -- Extensionality axiom: used sparingly, uses a back door in the type
   => system
50 -- Idris cannot rewrite types using axioms so this must be avoided at
   => all costs
51 public export
52 extensionality : {f, g : a -> b} -> ((x : a) -> f x = g x) -> f = g
53 extensionality {f} {g} prf = believe_me ()

```

### Verified Functors and Applicatives: VFunctor.idr

```

1 module Category.VFunctor
2
3 import Category.Morphism
4 import Data.Vect
5
6 %default total
7 %hide Applicative
8 %hide (<*>)
9 %hide (<$>)
10
11 infixl 4 <*>
12 infixl 4 <$>
13
14 -- Verified functors
15 -- Optics over functorial types can be verified in part using functor
   => laws
16 public export
17 interface VFunctor (f : Type -> Type) where
18   -- fmap maps functions
19   fmap : (a -> b) -> (f a -> f b)
20   -- fmap respects identity, F(id) = id
21   fid : (x : f a)
22     -> fmap (\x => x) x = x
23   -- fmap respects composition, F(g . h) = F(g) . F(h)
24   fcomp : (x : f a) -> (g : b -> c) -> (h : a -> b)
25     -> fmap (g . h) x = (fmap g . fmap h) x
26   -- Infix alias for fmap
27   (<$>) : (a -> b) -> (f a -> f b)
28   f <$> x = fmap f x
29   infixSame : (g : a -> b) -> (x : f a) -> fmap g x = g <$> x
30

```

```

31 -- Verified applicative functors
32 public export
33 interface VFunctor f => VApplicative (f : Type -> Type) where
34   -- pure (aka return,  $\Box$ )
35   ret : a -> f a
36   -- ap
37   (<*>) : f (a -> b) -> (f a -> f b)
38   -- Identity law, pure id <*> v = v
39   aid : (v : f a) -> ret (\x => x) <*> v = v
40   -- Homomorphism law, pure g <*> pure x = pure (g x)
41   ahom : (g : a -> b) -> (x : a)
42     -> ret g <*> ret x = ret (g x)
43   -- Interchange law, u <*> pure y = pure ($ y) <*> u
44   aint : (u : f (a -> b)) -> (y : a)
45     -> u <*> ret y = ret ($ y) <*> u
46   -- Composition law, ((pure (.) <*> u) <*> v) <*> w = u <*> (v <*> w)
47   acomp : (u : f (b -> c)) -> (v : f (a -> b)) -> (w : f a)
48     -> ((ret (.) <*> u) <*> v) <*> w = u <*> (v <*> w)
49
50 -- Lists are functors
51 public export
52 implementation VFunctor List where
53   -- fmap for lists is map
54   fmap f [] = []
55   fmap f (x::xs) = f x :: fmap f xs
56   fid [] = Refl
57   fid (x::xs) = cong (x::) (fid xs)
58   fcomp [] g h = Refl
59   fcomp (x::xs) g h = cong (g (h x) ::) (fcomp xs g h)
60   infixSame f x = Refl
61
62 -- forall xs. xs ++ [] = xs
63 public export
64 nilRightId : (xs : List a) -> xs ++ [] = xs
65 nilRightId [] = Refl
66 nilRightId (x::xs) =
67   let iH = nilRightId xs
68   in rewrite iH in Refl
69
70 -- List concatenation is associative
71 public export
72 concatAssoc : (xs, ys, zs : List a) -> xs ++ (ys ++ zs) = (xs ++ ys) ++
  ↪  zs
73 concatAssoc [] ys zs = Refl
74 concatAssoc (x::xs) ys zs = cong (x::) (concatAssoc xs ys zs)
75
76 -- Lists are applicative functors
77 public export

```

```

78 implementation VApplicative List where
79   -- pure makes a singleton list
80   ret = (::[])
81   -- ap applies a list of functions to a list of arguments
82   -- [f1, ..., fn] <*> [x1, ..., xn] = [f1 x1, ..., f1 xn, f2 x1, ...,
83     ↪ fn xn]
84   (<*>) [] xs = []
85   (<*>) (f::fs) xs = fmap f xs ++ (fs <*> xs)
86   -- Laws
87   aid [] = Refl
88   aid (x::xs) =
89     let iH = aid xs
90     shed : (fmap (\x => x) xs = xs) = fid xs
91     prf : ((fmap (\y => y) xs ++ []) = xs)
92     prf = rewrite shed in rewrite nilRightId xs in Refl
93   in cong (x::) prf
94   ahom g x = Refl
95   aint [] y = Refl
96   aint (u::us) y =
97     let iH = aint us y
98     in cong (u y::) iH
99   acomp us vs ws =
100     let elimNil : (((fmap (.) us ++ []) <*> vs) <*> ws = ((fmap (.) us)
101       ↪ <*> vs) <*> ws)
102     elimNil = cong (\x => (x <*> vs) <*> ws) (nilRightId (fmap (.)
103       ↪ us))
104   in rewrite elimNil in case us of
105     -- Goal: ((fmap (.) us) <*> vs) <*> ws = us <*> (vs <*> ws)
106     [] => Refl
107     (u::us') => let iH = acomp us' vs ws in let
108       l1 : List (a -> c)
109       l1 = fmap ((.) u) vs
110       l2 : List (a -> c)
111       l2 = (fmap (.) us') <*> vs
112       step : ((l1 ++ l2) <*> ws = (l1 <*> ws) ++ (l2 <*> ws))
113       step = concatDist l1 l2 ws
114       elimNil2 : (fmap u (vs <*> ws) ++ (<*>) ((fmap (.) us' ++ []))
115         ↪ <*> vs) ws = fmap u (vs <*> ws) ++ (((fmap (.) us') <*> vs)
116         ↪ <*> ws))
117       elimNil2 = cong (\x => fmap u (vs <*> ws) ++ (<*>) (x <*> vs)
118         ↪ ws) (nilRightId (fmap (.) us'))
119       prf : ((l1 ++ l2) <*> ws = fmap u (vs <*> ws) ++ (us' <*> (vs
120         ↪ <*> ws)))
121       prf = rewrite step in rewrite sym iH in rewrite elimNil2 in
122         cong (++ (((fmap (.) us') <*> vs) <*> ws)) (
123           -- Goal: ((fmap ((.) u) vs) <*> ws = fmap u (vs <*> ws))
124           case vs of
125             [] => Refl

```

```

119         (v::vs') => let
120             iH2 = acomp us' vs' ws
121             step2 : ((<*>) (fmap ((.) u) vs') ws = fmap u (vs' <*>
122                 ↪ ws))
123             step2 = apLemma u vs' ws
124             step3 : (fmap (u . v) ws ++ (<*>) (fmap ((.) u) vs') ws
125                 ↪ = fmap (u . v) ws ++ fmap u (vs' <*> ws))
126             step3 = cong (fmap (u . v) ws ++ step2)
127             step4 : (fmap (u . v) ws ++ fmap u (vs' <*> ws) = fmap
128                 ↪ u (fmap v ws) ++ fmap u (vs' <*> ws))
129             step4 = rewrite fcomp ws u v in Refl
130             step5 : (fmap u (fmap v ws) ++ fmap u (vs' <*> ws) =
131                 ↪ fmap u (fmap v ws ++ (vs' <*> ws)))
132             step5 = fmapHom u (fmap v ws) (vs' <*> ws)
133             final : (fmap (u . v) ws ++ ((fmap ((.) u) vs') <*> ws)
134                 ↪ = fmap u (fmap v ws ++ (vs' <*> ws)))
135             final = (step3 `trans` step4) `trans` step5
136             in final
137         )
138     in prf
139 where
140     -- Lemmas
141     -- Empty xs gives empty fs <*> xs
142     apRightNil : (fs : List (p -> q)) -> fs <*> [] = []
143     apRightNil [] = Refl
144     apRightNil (f::fs) = apRightNil fs
145     -- (<*>) distributes over (++)
146     concatDist : (as, bs : List (p -> q)) -> (xs : List p)
147         -> (as ++ bs) <*> xs = (as <*> xs) ++ (bs <*> xs)
148     concatDist [] bs xs = Refl
149     concatDist (a::as) bs xs = rewrite concatDist as bs xs in
150         concatAssoc (fmap a xs) (as <*> xs) (bs <*> xs)
151     -- fmap is a monoid homomorphism over the (List a, (++) , [])
152     ↪ monoid
153     fmapHom : (m : p -> q) -> (as, bs : List p)
154         -> fmap m as ++ fmap m bs = fmap m (as ++ bs)
155     fmapHom m [] bs = Refl
156     fmapHom m (a::as) bs = rewrite fmapHom m as bs in Refl
157     -- Function composition can be done before or after (<*>)
158     apLemma : (m : q -> r) -> (as : List (p -> q)) -> (bs : List p)
159         -> ((fmap ((.) m) as) <*> bs = fmap m (as <*> bs))
160     apLemma m [] bs = Refl
161     apLemma m (a::as) bs =
162         let iH = apLemma m as bs
163         in rewrite sym (fmapHom m (fmap a bs) (as <*> bs))
164         in rewrite sym iH
165         in rewrite fcomp bs m a
166         in Refl

```

```

161
162 -- Maybe is an applicative functor
163 public export
164 implementation VFunctor Maybe where
165     -- fmap maps over Just values
166     fmap f (Just x) = Just (f x)
167     fmap f Nothing = Nothing
168     fid (Just x) = Refl
169     fid Nothing = Refl
170     fcomp (Just x) g h = Refl
171     fcomp Nothing g h = Refl
172     infixSame f x = Refl
173
174 public export
175 implementation VApplicative Maybe where
176     ret = Just
177     -- ap returns a Just value iff it's possible to do so
178     (<*>) (Just f) (Just x) = Just (f x)
179     (<*>) _ _ = Nothing
180     aid (Just x) = Refl
181     aid Nothing = Refl
182     ahom g x = Refl
183     aint (Just f) y = Refl
184     aint Nothing y = Refl
185     acomp (Just u) (Just v) (Just w) = Refl
186     acomp Nothing _ _ = Refl
187     acomp (Just u) Nothing _ = Refl
188     acomp (Just u) (Just v) Nothing = Refl
189
190 -- Either a (partially applied sum type) is an applicative functor
191 -- over the second type variable
192 public export
193 implementation {a:Type} -> VFunctor (Either a) where
194     fmap f (Left x) = Left x
195     fmap f (Right x) = Right (f x)
196     fid (Left x) = Refl
197     fid (Right x) = Refl
198     fcomp (Left x) g h = Refl
199     fcomp (Right x) g h = Refl
200     infixSame f x = Refl
201
202 public export
203 implementation {a:Type} -> VApplicative (Either a) where
204     ret = Right
205     -- same as VApplicative Maybe, Left x is treated as Nothing and Right
206     --   ↪ x
207     -- as Just x
208     (<*>) (Right f) (Right x) = Right (f x)

```

```

208     (<*>) (Left x) y = Left x
209     (<*>) _ (Left x) = Left x
210     aid (Left x) = Refl
211     aid (Right x) = Refl
212     ahom g x = Refl
213     aint (Left x) y = Refl
214     aint (Right x) y = Refl
215     acomp (Right u) (Right v) (Right w) = Refl
216     acomp (Left _) _ _ = Refl
217     acomp (Right u) (Left x) _ = Refl
218     acomp (Right u) (Right v) (Left x) = Refl
219
220     -- Partially applied product type is a functor
221     -- over the second type variable
222     -- (a,) is only an applicative if a is a monoid (omitted)
223     public export
224     implementation {a:Type} -> VFunctor (a,) where
225         fmap f (x, y) = (x, f y)
226         fid (x, y) = Refl
227         fcomp (x, y) g h = Refl
228         infixSame f x = Refl
229
230     -- Morphism a = Hom(a, -) is an applicative functor,
231     -- the covariant Hom functor
232     public export
233     implementation {a:Type} -> VFunctor (Morphism a) where
234         -- fmap is function composition
235         -- The Mor wrapper is only present to help Idris unify types in
236         -- proofs
237         fmap f (Mor g) = Mor (f . g)
238         fid (Mor f) = cong Mor (sym (ext f))
239         fcomp (Mor f) g h = Refl
240         infixSame f x = Refl
241
242     public export
243     implementation {a:Type} -> VApplicative (Morphism a) where
244         ret x = Mor (const x)
245         (<*>) (Mor f) (Mor g) = Mor (\x => f x (g x))
246         aid (Mor x) = Refl
247         ahom g x = Refl
248         aint (Mor f) y = Refl
249         acomp (Mor u) (Mor v) (Mor w) = Refl
250
251     plusZeroRightId : (n : Nat) -> n + 0 = n
252     plusZeroRightId Z = Refl
253     plusZeroRightId (S n) = rewrite plusZeroRightId n in Refl
254
255     vectPlusZero : {n : Nat} -> Vect (plus n 0) a -> Vect n a

```



```

255 vectPlusZero xs = replace {p = \prf => Vect prf a} (plusZeroRightId n)
    ↪ xs
256
257 -- As with lists, length indexed vectors are functors
258 public export
259 implementation {n:Nat} -> VFunctor (Vect n) where
260     fmap f [] = []
261     fmap f (x::xs) = f x :: fmap f xs
262     fid [] = Refl
263     fid (x::xs) = cong (x::) (fid xs)
264     fcomp [] g h = Refl
265     fcomp (x::xs) g h = cong (g (h x) ::) (fcomp xs g h)
266     infixSame f x = Refl
267
268 -- Binary trees are functors
269 public export
270 data BTree : Type -> Type where
271     Null : BTree a
272     Node : BTree a -> a -> BTree a -> BTree a
273
274 public export
275 implementation VFunctor BTree where
276     -- fmap maps f recursively over the values in every node
277     fmap f Null = Null
278     fmap f (Node l x r) = Node (fmap f l) (f x) (fmap f r)
279     fid Null = Refl
280     fid (Node l x r) =
281         let iH1 = fid l
282             iH2 = fid r
283         in rewrite iH1
284         in rewrite iH2
285         in Refl
286     fcomp Null g h = Refl
287     fcomp (Node l x r) g h =
288         let iH1 = fcomp l g h
289             iH2 = fcomp r g h
290         in rewrite iH1
291         in rewrite iH2
292         in Refl
293     infixSame f x = Refl
294
295 -- Rose trees are functors
296 public export
297 data RTree : Type -> Type where
298     Leaf : a -> RTree a
299     Branch : List (RTree a) -> RTree a
300
301 -- These are for VFunctor RTree but had to be pulled out so `branches`

```

```

302 -- could be used in a proof about fmap as well as fmap
303 mutual
304   branches : (a -> b) -> List (RTree a) -> List (RTree b)
305   branches f [] = []
306   branches f (b::bs) = fmapRTree f b :: branches f bs
307
308   fmapRTree : (a -> b) -> (RTree a) -> (RTree b)
309   fmapRTree f (Leaf x) = Leaf (f x)
310   fmapRTree f (Branch bs) = Branch (branches f bs)
311
312 public export
313 implementation VFunctor RTree where
314   fmap = fmapRTree
315   fid (Leaf x) = Refl
316   fid (Branch bs) = cong Branch (prf bs) where
317     prf : (bs : List (RTree a)) -> branches (\x => x) bs = bs
318     prf [] = Refl
319     prf (b::bs) = rewrite prf bs in cong (::bs) (fid b)
320   fcomp (Leaf x) g h = Refl
321   fcomp (Branch bs) g h = cong Branch (prf bs g h) where
322     prf : (bs : List (RTree a)) -> (g : b -> c) -> (h : a -> b)
323         -> (branches (g . h) bs = branches g (branches h bs))
324     prf [] g h = Refl
325     prf (b::bs) g h = rewrite prf bs g h
326                       in cong (:: branches g (branches h bs)) (fcomp b g h)
327   infixSame f x = Refl

```

## Verified Profunctors: VProfunctor.idr

```

1  module Category.VProfunctor
2
3  import Category.VFunctor
4  import Category.Morphism
5
6  %default total
7  %hide Applicative
8
9  -- Verified profunctors
10 public export
11 interface VProfunctor (p : Type -> Type -> Type) where
12   -- dimap maps two morphisms over a profunctor
13   -- p(a,-) is a covariant functor, p(-,a) is contravariant
14   dimap : (a -> b) -> (c -> d) -> p b c -> p a d
15
16   -- Identity law, dimap id id = id
17   pid : {a, b : Type} -> (x : p a b) -> dimap (\x => x) (\x => x) x = x
18   -- Composition law, dimap (f' . f) (g . g') = dimap f g . dimap f' g'
19   pcomp

```

```

20     : {a, b, c, d, e, t : Type}
21     -> (x : p a b)
22     -> (f' : c -> a) -> (f : d -> c)
23     -> (g : e -> t) -> (g' : b -> e)
24     -> dimap (f' . f) (g . g') x = (dimap f g . dimap f' g') x
25
26 -- Profunctors for product and sum types, and monoidal profunctors
27
28 -- Cartesianly strong profunctors preserve product types
29 public export
30 interface VProfunctor p => Cartesian p where
31     first : p a b -> p (a, c) (b, c)
32     second : p a b -> p (c, a) (c, b)
33
34 -- Co-Cartesianly strong profunctors preserve sum types
35 public export
36 interface VProfunctor p => Cocartesian p where
37     left : p a b -> p (Either a c) (Either b c)
38     right : p a b -> p (Either c a) (Either c b)
39
40 -- Profunctors with monoid object structure
41 public export
42 interface VProfunctor p => Monoidal p where
43     par : p a b -> p c d -> p (a, c) (b, d)
44     empty : p () ()
45
46 -- Profunctor implementations
47
48 -- Hom(-,-) profunctor, the canonical profunctor
49 public export
50 implementation VProfunctor Morphism where
51     dimap f g (Mor h) = Mor (g . h . f)
52     pid (Mor f) = cong Mor (sym (ext f))
53     pcomp (Mor x) f' f g g' = Refl
54
55 public export
56 implementation Cartesian Morphism where
57     first (Mor f) = Mor (\(a, c) => (f a, c))
58     second (Mor f) = Mor (\(c, a) => (c, f a))
59
60 public export
61 implementation Cocartesian Morphism where
62     left (Mor f) = Mor (\case
63         Left a => Left (f a)
64         Right c => Right c)
65     right (Mor f) = Mor (\case
66         Left c => Left c
67         Right a => Right (f a))

```

```

68
69 public export
70 implementation Monoidal Morphism where
71   par (Mor f) (Mor g) = Mor (\(x, y) => (f x, g y))
72   empty = Mor (const ())
73
74   -- Hom profunctor in the Kleisli category
75   -- This is the category of monadic types `m a` with Kleisli composition
76   -- f . g = \x => join (f (g x)), where join : m (m a) -> m a
77   -- We only require a functor for convenience
78 public export
79 implementation {k : Type -> Type} -> VFunctor k => VProfunctor
80   ↪ (KleisliMorphism k) where
81     dimap f g (Kleisli h) = Kleisli (fmap g . h . f)
82     -- This proof reduces to `fmap (\x => x) . f = f` for `f : a -> k b`
83     -- We can't make `fid` intensional, ie `fid : fmap (\x => x) = id`,
84     -- because we need something to pattern match on to prove fid, so we
85     ↪ must use
86     -- extensionality here
87     pid (Kleisli f) = cong Kleisli (extensionality (\x => fid (f x)))
88     pcomp (Kleisli u) f' f g g' = cong Kleisli (extensionality (\x =>
89       fcomp (u (f' (f x))) g g'))
90
91 public export
92 implementation {k : Type -> Type} -> VApplicative k => Cocartesian
93   ↪ (KleisliMorphism k) where
94     left (Kleisli f) = Kleisli (either (fmap Left . f) (ret . Right))
95     right (Kleisli f) = Kleisli (either (ret . Left) (fmap Right . f))
96
97   -- Const profunctor, Const r a is isomorphic to Hom((), a)
98   -- This profunctor allows us to use our optics as constructors
99   -- eg: op {p=Const} (MkConst 3) == MkConst (Just 3)
100 public export
101 record Const r a where
102   constructor MkConst -- MkConst : a -> Const r a
103   unConst : a -- unConst : Const r a -> a
104
105 public export
106 implementation VProfunctor Const where
107   dimap f g (MkConst x) = MkConst (g x)
108   pid (MkConst x) = Refl
109   pcomp (MkConst x) f' f g g' = Refl
110
111 public export
112 implementation Cocartesian Const where
113   left (MkConst x) = MkConst (Left x)
114   right (MkConst x) = MkConst (Right x)

```

```

113 public export
114 implementation Monoidal Const where
115   par (MkConst x) (MkConst y) = MkConst (x, y)
116   empty = MkConst ()
117
118 -- `Forget r` profunctor
119 -- Allows us to use our profunctor optics as getters
120 -- eg: unForget ( $\pi_1$  {p=Forget Int} (MkForget (\x => x))) (3, True) == 3
121 -- Inspired by PureScript's profunctor-lenses:
122 -- https://github.com/purescript-contrib/purescript-profunctor-lenses/
123 public export
124 record Forget r a b where
125   constructor MkForget -- MkForget : (a -> r) -> Forget r a b
126   unForget : a -> r -- unForget : Forget r a b -> (a -> r)
127
128 public export
129 implementation {r : Type} -> VProfunctor (Forget r) where
130   dimap f g (MkForget h) = MkForget (h . f)
131   pid (MkForget x) = Refl
132   pcomp (MkForget x) f' f g g' = Refl
133
134 public export
135 implementation {r : Type} -> Cartesian (Forget r) where
136   first (MkForget f) = MkForget (\(x, y) => f x)
137   second (MkForget f) = MkForget (\(x, y) => f y)

```

## Simple Optics: PrimitiveOptics.idr

```

1 module Primitive.PrimitiveOptics
2
3 %default total
4
5 -- Primitive optics
6 -- Simpler to write than profunctor optics but they don't compose well
7 -- Solution: write primitive optics and map them to profunctor optics
8
9 public export
10 data PrimLens : Type -> Type -> Type -> Type -> Type where
11   MkPrimLens
12     : (view : s -> a)
13     -> (update : (b, s) -> t)
14     -> PrimLens a b s t
15
16 public export
17 data PrimPrism : Type -> Type -> Type -> Type -> Type where
18   MkPrimPrism
19     : (match : s -> Either t a)
20     -> (build : b -> t)

```

```

21     -> PrimPrism a b s t
22
23 public export
24 data PrimAdapter : Type -> Type -> Type -> Type -> Type where
25   MkPrimAdapter
26     : (from : s -> a)
27     -> (to : b -> t)
28     -> PrimAdapter a b s t
29
30 -- Examples of simple optics
31
32 -- Product left/right projection lens
33  $\pi_1$  : PrimLens a b (a, c) (b, c)
34  $\pi_1$  = MkPrimLens fst update where
35   update : (b, (a, c)) -> (b, c)
36   update (x', (x, y)) = (x', y)
37  $\pi_2$  : PrimLens a b (c, a) (c, b)
38  $\pi_2$  = MkPrimLens snd update where
39   update : (b, (c, a)) -> (c, b)
40   update (x', (y, x)) = (y, x')
41
42 -- Sign of an integer lens
43 sgn : PrimLens Bool Bool Integer Integer
44 sgn = MkPrimLens signum chsgn where
45   signum : Integer -> Bool
46   signum x = x >= 0
47   chsgn : (Bool, Integer) -> Integer
48   chsgn (True, x) = abs x
49   chsgn (False, x) = -abs x
50
51 -- Maybe prism
52 public export
53 op' : PrimPrism a b (Maybe a) (Maybe b)
54 op' = MkPrimPrism match build where
55   match : Maybe a -> Either (Maybe b) a
56   match (Just x) = Right x
57   match Nothing = Left Nothing
58   build : b -> Maybe b
59   build = Just
60
61 -- Adapter for the isomorphism  $(A \times B) \times C = A \times (B \times C)$ 
62 prodAssoc : PrimAdapter ((a,b),c) ((a',b'),c') (a,(b,c)) (a',(b',c'))
63 prodAssoc = MkPrimAdapter (\(x,(y,z)) => ((x,y),z)) (\((x,y),z) =>
  ↪ (x,(y,z)))

```

## van Laarhoven Optics: LaarhovenOptics.idr

```
1 module Primitive.LaarhovenOptics
2
3 import Category.VFunctor
4
5 -- van Laarhoven lens type
6 public export
7 LaarhovenLens : {f : Type -> Type} -> VFunctor f => Type -> Type ->
  ↳ Type
8 LaarhovenLens a s = (a -> f a) -> (s -> f s)
9
10 -- Left product projection
11 public export
12 laarhovenProj : {f : Type -> Type} -> VFunctor f => LaarhovenLens {f=f}
  ↳ a (a,b)
13 laarhovenProj g (x, y) = fmap (,y) (g x)
14
15 -- The Const functor turns van Laarhoven optics into getters
16 -- Note this Const stores one of the first type and the Const in
  ↳ VProfunctor
17 -- stores one of the second type
18 record Const a b where
19   constructor MkConst
20   unConst : a
21
22 implementation {a : Type} -> VFunctor (Const a) where
23   fmap f (MkConst x) = MkConst x
24   fid (MkConst x) = Refl
25   fcomp (MkConst x) g h = Refl
26   infixSame g (MkConst x) = Refl
27
28 -- Identity functor
29 public export
30 record Id a where
31   constructor MkId
32   unId : a
33
34 -- The identity functor turns van Laarhoven optics into update
  ↳ functions
35 public export
36 implementation VFunctor Id where
37   fmap f (MkId x) = MkId (f x)
38   fid (MkId x) = Refl
39   fcomp (MkId x) g h = Refl
40   infixSame g (MkId x) = Refl
41
42 -- Useful combinators
```

```

43 public export
44 view' : LaarhovenLens {f=Const} a s -> (s -> a)
45 view' optic structure = unConst $
46   optic (\x => MkConst x) structure
47
48 public export
49 update' : LaarhovenLens {f=Id} a s -> ((a, s) -> s)
50 update' optic (field, structure) = unId $
51   optic (\x => MkId field) structure

```

## Profunctor Optics: Main.idr

```

1 module Main
2
3 import Category.VProfunctor
4 import Category.VFunctor
5 import Category.Morphism
6 import Primitive.PrimitiveOptics
7 import Primitive.LaarhovenOptics
8 import Data.Vect
9
10 %default total
11 %hide Prelude.Interfaces.<*>
12 %hide Prelude.Interfaces.<$>
13
14 infixr 0 ~>
15
16 -- Profunctor optic types
17
18 Optic : (Type -> Type -> Type) -> Type -> Type -> Type -> Type -> Type
19 Optic p a b s t = p a b -> p s t
20
21 Adapter : Type -> Type -> Type -> Type -> Type
22 Adapter a b s t = {p : Type -> Type -> Type} -> VProfunctor p => Optic
   -> p a b s t
23
24 Lens : Type -> Type -> Type -> Type -> Type
25 Lens a b s t = {p : Type -> Type -> Type} -> Cartesian p => Optic p a b
   -> s t
26
27 Prism : Type -> Type -> Type -> Type -> Type
28 Prism a b s t = {p : Type -> Type -> Type} -> Cocartesian p => Optic p
   -> a b s t
29
30 LensPrism : Type -> Type -> Type -> Type -> Type
31 LensPrism a b s t = {p : Type -> Type -> Type}
32   -> (Cartesian p, Cocartesian p)
33   => Optic p a b s t

```



```

34
35 Traversal : Type -> Type -> Type -> Type -> Type
36 Traversal a b s t = {p : Type -> Type -> Type}
37   -> (Cartesian p, Cocartesian p, Monoidal p)
38   => Optic p a b s t
39
40 -- Helpful combinators
41
42 -- `Forget r` profunctor optics operate as getters
43 view : {a : Type} -> Lens a b s t -> s -> a
44 view optic x = unForget (optic {p=Forget a} (MkForget (\x => x))) x
45
46 -- Morphism profunctor optics operate as setters
47 update : Optic Morphism a b s t -> (a -> b) -> (s -> t)
48 update optic f x = applyMor (optic (Mor f)) x
49
50 -- Const profunctor optics recovers sum type constructors
51 build : Prism a b s t -> b -> t
52 build optic x = unConst (optic {p=Const} (MkConst x))
53
54 -- Product type optics
55
56 --  $\pi_1 : \{p : \text{Type} \rightarrow \text{Type} \rightarrow \text{Type}\} \rightarrow \text{Cartesian } p \Rightarrow p \ a \ b \rightarrow p \ (a, \ c)$ 
57    $\hookrightarrow \ (b, \ c)$ 
58  $\pi_1$  : Lens a b (a, c) (b, c)
59  $\pi_1$  = first
60
61  $\pi_2$  : Lens a b (c, a) (c, b)
62  $\pi_2$  = second
63
64 -- Optional type optics
65
66 --  $op : \{p : \text{Type} \rightarrow \text{Type} \rightarrow \text{Type}\} \rightarrow \text{Cocartesian } p \Rightarrow p \ a \ b \rightarrow p$ 
67    $\hookrightarrow \ (\text{Maybe } a) \ (\text{Maybe } b)$ 
68 op : Prism a b (Maybe a) (Maybe b)
69 op = dimap (maybe (Left Nothing) Right) (either id Just) . right
70
71 -- Sum/coproduct type optics
72
73 leftP : Prism a b (Either a c) (Either b c)
74 leftP = left
75
76 rightP : Prism a b (Either c a) (Either c b)
77 rightP = right
78
79 -- Example of composition of optics
80
81 op_ $\pi_1$  : LensPrism a b (Maybe (a, c)) (Maybe (b, c))

```

```

80 op_π1 = op . π1
81
82 -- Map primitive optics to profunctor optics
83
84 prismPrimToPro : PrimPrism a b s t -> Prism a b s t
85 prismPrimToPro (MkPrimPrism m b) = dimap m (either id b) . right
86
87 -- Complex data structures
88
89 -- This type is from van Laarhoven
90   ↳ (https://twanvl.nl/blog/haskell/non-regular1)
91 -- FunList a b t is isomorphic to ∃n. an × (bn -> t)
92 -- which is equivalent to the type of a traversable (Pickering et. al.
93   ↳ 2018)
94 -- It allows us to write optics for lists and trees
95 -- This is ported from the Haskell code from Pickering et. al. 2018
96 data FunList : Type -> Type -> Type -> Type where
97   Done : t -> FunList a b t
98   More : a -> FunList a b (b -> t) -> FunList a b t
99
100 out : FunList a b t -> Either t (a, FunList a b (b -> t))
101 out (Done t) = Left t
102 out (More x l) = Right (x, l)
103
104 inn : Either t (a, FunList a b (b -> t)) -> FunList a b t
105 inn (Left t) = Done t
106 inn (Right (x, l)) = More x l
107
108 implementation {a : Type} -> {b : Type} -> VFunctor (FunList a b) where
109   fmap f (Done t) = Done (f t)
110   fmap f (More x l) = More x (fmap f l)
111   fid (Done t) = Refl
112   fid (More x l) = cong (More x) (fid l)
113   fcomp (Done t) g h = Refl
114   fcomp (More x l) g h = cong (More x) (fcomp l (g .) (h .))
115   infixSame f x = Refl
116
117 implementation {a : Type} -> {b : Type} -> VApplicative (FunList a b)
118   ↳ where
119   ret = Done
120   Done f <*> l = fmap f l
121   More x l <*> l2 = assert_total More x (fmap flip l <*> l2)
122   aid (Done t) = Refl
123   aid (More x l) = cong (More x) (aid l)
124   ahom g x = Refl
125   aint u y = believe_me () -- todo
126   acomp u v w = believe_me ()

```

```

125 single : a -> FunList a b b
126 single x = More x (Done id)
127
128 fuse : FunList b b t -> t
129 fuse (Done t) = t
130 fuse (More x l) = fuse l x
131
132 traverse : {p : Type -> Type -> Type} -> (Cocartesian p, Monoidal p)
133   => p a b
134   -> p (FunList a c t) (FunList b c t)
135 traverse k = assert_total dimap out inn (right (par k (traverse k)))
136
137 makeTraversal : (s -> FunList a b t) -> Traversal a b s t
138 makeTraversal h k = dimap h fuse (traverse k)
139
140 -- Binary tree traversals
141
142 inorder' : {f : Type -> Type} -> VApplicative f
143   => (a -> f b)
144   -> BTree a -> f (BTree b)
145 inorder' m Null = ret Null
146 inorder' m (Node l x r) = Node <$> inorder' m l <*> m x <*> inorder' m
147   ↪ r
148
149 inorder : {a, b : Type} -> Traversal a b (BTree a) (BTree b)
150 inorder = makeTraversal (inorder' single)
151
152 preorder' : {f : Type -> Type} -> VApplicative f
153   => (a -> f b)
154   -> BTree a -> f (BTree b)
155 preorder' m Null = ret Null
156 preorder' m (Node l x r) =
157   (\\mid, left, right => Node left mid right) <$>
158   m x <*> preorder' m l <*> preorder' m r
159
160 preorder : {a, b : Type} -> Traversal a b (BTree a) (BTree b)
161 preorder = makeTraversal (preorder' single)
162
163 postorder' : {f : Type -> Type} -> VApplicative f
164   => (a -> f b)
165   -> BTree a -> f (BTree b)
166 postorder' m Null = ret Null
167 postorder' m (Node l x r) =
168   (\\left, right, mid => Node left mid right) <$>
169   postorder' m l <*> postorder' m r <*> m x
170
171 postorder : {a, b : Type} -> Traversal a b (BTree a) (BTree b)
172 postorder = makeTraversal (postorder' single)

```

```

172
173 -- List traversals
174
175 listTraverse' : {f : Type -> Type} -> VApplicative f
176   => (a -> f b)
177   -> List a -> f (List b)
178 listTraverse' g [] = ret []
179 listTraverse' g (x::xs) = (::) <$> g x <*> listTraverse' g xs
180
181 listTraverse : {a, b : Type} -> Traversal a b (List a) (List b)
182 listTraverse = makeTraversal (listTraverse' single)
183
184 -- PrimPrism a b forms a Cocartesian profunctor
185
186 -- Definitions and lemmas from the Either bifunctor for `VProfunctor
187   ⇨ (PrimPrism a b)`
188 bimapEither : (a -> c) -> (b -> d) -> Either a b -> Either c d
189 bimapEither f g (Left x) = Left (f x)
190 bimapEither f g (Right x) = Right (g x)
191
192 bimapId : (z : Either a b) -> bimapEither (\x => x) (\x => x) z = z
193 bimapId (Left y) = Refl
194 bimapId (Right y) = Refl
195
196 bimapLemma : (g : e -> t) -> (g' : b -> e) -> (x' : Either b a)
197   -> bimapEither (g . g') (\x => x) x' = bimapEither g (\x => x)
198   ⇨ (bimapEither g' (\x => x) x')
199 bimapLemma g g' (Left x) = Refl
200 bimapLemma g g' (Right x) = Refl
201
202 public export
203 implementation {a : Type} -> {b : Type} -> VProfunctor (PrimPrism a b)
204   ⇨ where
205     dimap f g (MkPrimPrism m b) = MkPrimPrism (bimapEither g id . m . f)
206     ⇨ (g . b)
207     pid (MkPrimPrism m b) = cong (`MkPrimPrism` b)
208     (extensionality (\x => bimapId (m x)))
209     pcomp (MkPrimPrism m b) f' f g g' = cong (`MkPrimPrism` (\x => g (g'
210     ⇨ (b x))))
211     (extensionality (\x => bimapLemma g g' (m (f' (f x)))))
212
213 public export
214 implementation {a : Type} -> {b : Type} -> Cocartesian (PrimPrism a b)
215   ⇨ where
216     left (MkPrimPrism m b) = MkPrimPrism (either (bimapEither Left id .
217     ⇨ m) (Left . Right)) (Left . b)
218     right (MkPrimPrism m b) = MkPrimPrism (either (Left . Left)
219     ⇨ (bimapEither Right id . m)) (Right . b)

```

```

212
213 -- Unit tests (if these fail we get type errors)
214 -- These are provided as examples of how to use these profunctor optics
    ⇨ in practice
215
216 test1 : update (Main.op .  $\pi_1$ ) (\x => x * x) (Just (3, True)) = Just (9,
    ⇨ True)
217 test1 = Refl
218
219 test2 : view  $\pi_1$  (3, True) = 3
220 test2 = Refl
221
222 test3 : build Main.op 3 = Just 3
223 test3 = Refl
224
225 -- view  $\pi_1$  = fst (extensionally)
226 forgetLeftProjection : (x : r) -> (y : b)
227   -> fst (x, y) = view  $\pi_1$  (x, y)
228 forgetLeftProjection x y = Refl
229
230 -- build op = Just (extensionally)
231 constBuildsMaybe : (x : a)
232   -> Just x = build Main.op x
233 constBuildsMaybe x = Refl

```